

The Stata Journal (2012) **12**, Number 3, pp. 433–446

# Easy demand-system estimation with quaids

Brian P. Poi StataCorp LP College Station, TX bpoi@stata.com

Abstract. Previously, to fit an almost-ideal demand system in Stata, one would have to use the nlsur command and write a function evaluator program as described in [R] nlsur and Poi (2008, Stata Journal 8: 554–556). In this article, I introduce the command quaids, which obviates the need for any programming by the user. The command fits Deaton and Muellbauer's (1980b, American Economic Review 70: 312–326) original almost-ideal demand-system model as well as Banks, Blundell, and Lewbel's (1997, Review of Economics and Statistics 79: 527–539) quadratic variant. Demographic variables can also be included in the model. Postestimation tools calculate expenditure and price elasticities.

**Keywords:** st0268, quaids, almost-ideal demand system

### 1 Introduction

Demand-system estimation is a popular task in Stata. Questions arise regularly on the Statalist email discussion forum, and I frequently receive private requests for assistance from users. Previously, to fit an almost-ideal demand system (AIDS) in Stata, one would have to use the nlsur command and write a function evaluator program as described in [R] nlsur and Poi (2008). However, many users had difficulty adapting the function evaluator program to fit their specific applications. Moreover, adding demographic variables was difficult, and obtaining price and expenditure elasticities required even more programming.

The quaids command solves those shortcomings. quaids allows you to fit either the standard AIDS model of Deaton and Muellbauer (1980b) or the quadratic AIDS model of Banks, Blundell, and Lewbel (1997). Demographic variables can be specified and are incorporated using Ray's (1983) method. Postestimation commands allow you to compute expenditure elasticities as well as compensated and uncompensated price elasticities.

The rest of this article is organized as follows. Section 2 outlines the models and elasticity computations. Section 3 describes the quaids command and associated postestimation commands. Section 4 provides an example. Section 5 concludes the article.

# 2 Almost-ideal demand systems

In this section, I describe AIDS models and show how demographic variables are incorporated by the quaids command. The presentation here is brief; for an in-depth analysis of consumer behavior and demand-system analysis, see the classic monographs by Deaton and Muellbauer (1980a) and Mas-Colell, Whinston, and Green (1995, chap. 3).

## 2.1 Basics

We consider a consumer's demand for a set of k goods for which the consumer has budgeted m units of currency. For example, the k goods could represent different categories of food, and the amount to be spent on food, m, could be chosen based on a two-stage budgeting process. Alternatively, the k goods could represent broad categories like housing, clothing, food, utilities, and savings; m could represent household income.

The quadratic AIDS model of Banks, Blundell, and Lewbel (1997) is based on the indirect utility function

$$\ln V(\mathbf{p}, m) = \left[ \left\{ \frac{\ln m - \ln a(\mathbf{p})}{b(\mathbf{p})} \right\}^{-1} + \lambda(\mathbf{p}) \right]^{-1}$$
 (1)

where  $\ln a(\mathbf{p})$  is the transcendental logarithm function

$$\ln a(\mathbf{p}) = \alpha_0 + \sum_{i=1}^k \alpha_i \ln p_i + \frac{1}{2} \sum_{i=1}^k \sum_{j=1}^k \gamma_{ij} \ln p_i \ln p_k$$
 (2)

In this function,  $p_i$  is the price of good i for i = 1, ..., k;  $b(\mathbf{p})$  is the Cobb-Douglas price aggregator

$$b(\mathbf{p}) = \prod_{i=1}^{k} p_i^{\beta_i}$$

and

$$\lambda(\mathbf{p}) = \sum_{i=1}^{k} \lambda_i \ln p_i$$

Lowercase Greek letters other than  $\alpha_0$  represent parameters to be estimated. In principle,  $\alpha_0$  could be estimated jointly with the other parameters; in practice, it turns out to be rather difficult. The quaids command requires the user to specify a value for  $\alpha_0$ ; Deaton and Muellbauer (1980b), Banks, Blundell, and Lewbel (1997), and most others set  $\alpha_0$  to be slightly less than the lowest value of  $\ln m$  observed in the data.

Adding up, homogeneity, and Slutsky symmetry impose the requirements that

$$\sum_{i=1}^{k} \alpha_i = 1, \qquad \sum_{i=1}^{k} \beta_i = 0, \qquad \sum_{j=1}^{k} \gamma_{ij} = 0, \qquad \sum_{i=1}^{k} \lambda_i = 0, \quad \text{and} \quad \gamma_{ij} = \gamma_{ji}$$

The quaids command imposes these conditions automatically; the current implementation does not allow the user to fit unrestricted models and test whether these axioms of demand analysis hold.

Let  $q_i$  denote the quantity of good i consumed by a household, and define the expenditure share for good i as  $w_i = p_i q_i/m$ . Applying Roy's identity to (1), we obtain the expenditure share equation for good i:

$$w_i = \alpha_i + \sum_{j=1}^k \gamma_{ij} \ln p_j + \beta_i \ln \left\{ \frac{m}{a(\mathbf{p})} \right\} + \frac{\lambda_i}{b(\mathbf{p})} \left[ \ln \left\{ \frac{m}{a(\mathbf{p})} \right\} \right]^2, \qquad i = 1, \dots, k$$

When  $\lambda_i = 0$  for all i, the quadratic term in each expenditure share equation drops out, and we are left with Deaton and Muellbauer's (1980b) original AIDS model. Thus testing  $H_0: \lambda_i = 0 \,\forall i$  allows us to choose easily between the original AIDS and the quadratic AIDS models.

Consider the original AIDS model without the quadratic term:

$$w_i = \alpha_i + \sum_{j=1}^k \gamma_{ij} \ln p_j + \beta_i \ln \left\{ \frac{m}{a(\mathbf{p})} \right\}, \qquad i = 1, \dots, k$$

This set of expenditure share equations requires nonlinear estimation techniques because of the price index  $\ln a(\mathbf{p})$ . Deaton and Muellbauer (1980b, 316) suggest replacing that price index with the approximation  $\ln a(\mathbf{p}) \approx \sum_j w_j \ln p_j$ , resulting in a set of equations that can be fit by linear estimation techniques. Occasionally, I am asked how to fit those so-called "linearly approximated" AIDS models in Stata. However, given the advances in software in the last 30 years and Stata's nlsur command, fitting nonlinear systems nowadays is not much more difficult than fitting linear systems. Therefore, the quaids command currently uses the price index defined in (2) and does not offer the option of using alternative price indices.

# 2.2 Demographics

The quaids command incorporates demographics by using the scaling technique introduced by Ray (1983) and extended to the quadratic AIDS model by Poi (2002a). We use  $\mathbf{z}$  to represent a vector of s characteristics. In the simplest case,  $\mathbf{z}$  could be a scalar representing the number of people in a household. Let  $e^R(\mathbf{p}, u)$  denote the expenditure function of a reference household, where a reference household might be one that contains just a single adult.

<sup>1.</sup> This price index was first introduced by Sir Richard Stone.

Ray's method uses for each household an expenditure function of the form

$$e(\mathbf{p}, \mathbf{z}, u) = m_0(\mathbf{p}, \mathbf{z}, u) \times e^R(\mathbf{p}, u)$$

The function  $m_0(\mathbf{p}, \mathbf{z}, u)$  scales the expenditure function to account for the household characteristics. Ray further decomposes the scaling function as

$$m_0(\mathbf{p}, \mathbf{z}, u) = \overline{m}_0(\mathbf{z}) \times \phi(\mathbf{p}, \mathbf{z}, u)$$

The first term measures the increase in a household's expenditures as a function of  $\mathbf{z}$ , not controlling for any changes in consumption patterns; a household with four members will have higher expenditures than one with a single member, even ignoring that the composition of goods consumed may change. The second term controls for changes in relative prices and the actual goods consumed; a household with two adults and two infants will consume different goods than one comprising four adults.

Following Ray (1983), quaids parameterizes  $\overline{m}_0(\mathbf{z})$  as

$$\overline{m}_0(\mathbf{z}) = 1 + \boldsymbol{\rho}' \mathbf{z}$$

where  $\rho$  is a vector of parameters to be estimated. As in Poi (2002a), quaids parameterizes  $\phi(\mathbf{p}, \mathbf{z}, u)$  as

$$\ln \phi(\mathbf{p}, \mathbf{z}, u) = \frac{\prod_{j=1}^{k} p_j^{\beta_j} \left( \prod_{j=1}^{k} p_j^{\boldsymbol{\eta}_j' \mathbf{z}} - 1 \right)}{\frac{1}{u} - \sum_{j=1}^{k} \lambda_j \ln p_j}$$

This functional form may look a bit odd, but it has the distinct advantage of resulting in expenditure share equations that closely mimic their counterparts without demographics.  $\eta_j$  represents the jth column of  $s \times k$  parameter matrix  $\eta$ . The expenditure share equations take the form

$$w_i = \alpha_i + \sum_{j=1}^k \gamma_{ij} \ln p_j + (\beta_i + \boldsymbol{\eta}_i' \mathbf{z}) \ln \left\{ \frac{m}{\overline{m}_0(\mathbf{z}) a(\mathbf{p})} \right\} + \frac{\lambda_i}{b(\mathbf{p}) c(\mathbf{p}, \mathbf{z})} \left[ \ln \left\{ \frac{m}{\overline{m}_0(\mathbf{z}) a(\mathbf{p})} \right\} \right]^2$$

where

$$c(\mathbf{p}, \mathbf{z}) = \prod_{j=1}^{k} p_j^{\boldsymbol{\eta}_j' \mathbf{z}}$$

The adding-up condition requires that  $\sum_{j=1}^{k} \eta_{rj} = 0$  for r = 1, ..., s. If we set  $\lambda_i = 0$  for all i, we are left with the AIDS model with demographics used by Ray (1983).<sup>2</sup>

<sup>2.</sup> This is not the only way to incorporate demographics into the quadratic AIDS model using Ray's (1983) method. While writing this article, I became aware of an article by Blacklow, Nicholas, and Ray (2010) that used Ray's (1983) method to incorporate household composition variables in the quadratic AIDS model. Also see Pollak and Wales (1981), who discuss alternative ways of incorporating demographic variables by directly modifying the expenditure share equations.

#### 2.3 Elasticities

We present the formulas for the elasticities for the quadratic AIDS model with demographic variables. The formulas for elasticities for the standard AIDS model and models without demographics are nested within the more general variants.

Tedious but straightforward algebra shows that the uncompensated price elasticity of good i with respect to changes in the price of good j is

$$\epsilon_{ij} = -\delta_{ij} + \frac{1}{w_i} \left( \gamma_{ij} - \left[ \beta_i + \boldsymbol{\eta}_i' \mathbf{z} + \frac{2\lambda_i}{b(\mathbf{p}) c(\mathbf{p}, \mathbf{z})} \ln \left\{ \frac{m}{\overline{m}_0(\mathbf{z}) a(\mathbf{p})} \right\} \right] \times \left( \alpha_j + \sum_l \gamma_{jl} \ln p_l \right) - \frac{\left( \beta_j + \boldsymbol{\eta}_j' \mathbf{z} \right) \lambda_i}{b(\mathbf{p}) c(\mathbf{p}, \mathbf{z})} \left[ \ln \left\{ \frac{m}{\overline{m}_0(\mathbf{z}) a(\mathbf{p})} \right\} \right]^2 \right)$$

The expenditure (income) elasticity for good i is

$$\mu_i = 1 + \frac{1}{w_i} \left[ \beta_i + \boldsymbol{\eta}_i' \mathbf{z} + \frac{2\lambda_i}{b(\mathbf{p}) c(\mathbf{p}, \mathbf{z})} \ln \left\{ \frac{m}{\overline{m}_0(\mathbf{z}) a(\mathbf{p})} \right\} \right]$$
(3)

Compensated price elasticities are obtained from the Slutsky equation:  $\epsilon_{ij}^C = \epsilon_{ij} + \mu_i w_j$ .

#### 2.4 Estimation

We assume there is an additive zero-mean error term associated with each of the k expenditure share equations. We estimate the parameters by iterated feasible generalized nonlinear least-squares estimation via Stata's nlsur command with the ifgnls option. To avoid a singular error-covariance matrix, the quaids command automatically omits the last equation before calling nlsur; the estimator is invariant to which equation is dropped, so there is no option to make quaids drop some equation other than the last. The iterated feasible generalized nonlinear least-squares estimator is equivalent to the multivariate normal maximum-likelihood estimator for this class of problems.

The function evaluator program required by nlsur, named nlsur\_quaids.ado, is a wrapper command that calls a Mata routine to compute the predicted expenditure share equations. The elasticity computations are also performed by Mata routines. All of these routines are stored in the \_quaids\_\_utils.mata file, which accompanies this article.

# 3 The quaids command suite

Accompanying this article is a complete suite of commands for fitting AIDS models, obtaining forecasts, and calculating elasticities. These programs require Stata 12.1.

#### 3.1 Estimation command

The quaids command fits an AIDS model, with or without the quadratic expenditure term and with or without demographic variables.

```
quaids varlist_{expshares} [if] [in], anot(#)
 \{ \underbrace{\text{prices}(varlist_{prices}) | \underline{\text{Inprices}}(varlist_{lnprices})} \} 
 \{ \underbrace{\text{expenditure}(varname_{exp}) | \underline{\text{Inexpenditure}}(varname_{lnexp})} \} 
 [ \underline{\text{demographics}(varlist_{demo}) \text{ noquadratic } \underline{\text{nolog }} \text{ vce}(vcetype) \text{ } \underline{\text{level}}(\#) ]
```

 $varlist_{expshares}$  represents the variables containing the expenditure shares. To fit a system with k goods, specify k variable names. Do not omit one of the goods to avoid a singular error-covariance matrix during estimation; the quaids command handles that automatically. The expenditure share variables should sum to 1 for each observation; quaids exits with an error message if that is not true of the data.

Specify k price variables,  $varlist_{prices}$ , in the prices() option or specify k variables containing the natural logarithms of the prices of the k goods,  $varlist_{lnprices}$ , in the lnprices() option. Specify one or the other but not both. The price variables must appear in the same order as the expenditure share variables; quaids has no way to verify this, but results will be meaningless if the ordering is not preserved.

Specify the total expenditure variable,  $varname_{exp}$ , in the option expenditure() or specify the variable containing the natural logarithm of expenditure,  $varname_{lnexp}$ , in the option lnexpenditure(). Specify one or the other but not both. The expenditure variable should represent total expenditures on the goods that this demand system represents such that  $\sum_{j} p_{j}q_{j} = m$  for each observation in the dataset.

Specify the value for  $\alpha_0$  in option anot(). Specify optional demographic variables,  $varlist_{demo}$ , in option demographics(). Option noquadratic instructs quaids to omit the quadratic term. Option nolog suppresses the iteration log. Option level(#) sets the confidence level for confidence intervals.

vce(vcetype) controls the type of variance—covariance matrix calculated. vcetype may be gnr (the default), robust, cluster clustvar (where clustvar is the name of a clustering variable), bootstrap, or jackknife. This option works just as it does for nlsur; see [R] nlsur.

#### 3.2 Prediction

After fitting an AIDS model, use the predict command to obtain predicted values of the expenditure shares.

```
predict [type] \{stub* | newvar_1, ..., newvar_k\} [if] [in]
```

Specify stub immediately followed by an asterisk (\*), or specify k new variable names, where k is the number of goods in the demand system. If the user specifies, say, what\*, then the new variables created will be named what1, ..., whatk. By default, the predictions are computed for all observations in the dataset; specify if e(sample) to restrict computations to only those observations used during estimation.

#### 3.3 Elasticities

One is often more interested in the expenditure and price elasticities rather than in the parameters of the expenditure share equations per se. However, I suspect that many Stata users attempting to fit demand systems give up before computing the elasticities. There are at least two reasons. First, some users have had significant difficulty writing the requisite function evaluator program for use with nlsur, so writing a program to compute the elasticities or using the nlcom command to do so may be out of reach. Second, I often receive email questions asking for help fitting an AIDS model, but remarkably, few of those users ask about obtaining elasticities.

The quaids command includes three postestimation utilities for computing expenditure elasticities as well as compensated and uncompensated price elasticities. These commands allow you to compute the elasticities at the means of all the variables in the model or to compute the elasticities for each observation in the current dataset. You can compute elasticities for various subsets of the data by creating an artificial dataset containing the appropriate variable means. The example in the next section illustrates how to compute elasticities for "average" rural and urban households. The current implementation does not compute standard errors for the elasticity estimates.

All three of the estat commands accompanying this article will compute elasticities for each observation in the dataset and place the results in new variables. Alternatively, you may specify atmeans to have the commands first compute the means of the price, expenditure, and demographic variables, and then compute the elasticities at the means of those variables; with this option, the elasticities are returned as a vector or matrix. In both cases, by default, the estat commands compute the statistics for all the observations in the dataset unless you use the if or in modifiers to restrict the sample.

To compute the expenditure elasticities for individual observations in the dataset, you type

```
estat expenditure [type] \{stub* | newvar_1, ..., newvar_k\} [if] [in]
```

Similarly to the predict command discussed above, you may specify either a variable stub or k new variable names. The jth new variable will contain the expenditure elasticity for the jth good.

To compute the expenditure elasticities at the means of all the variables and have the result returned in a  $1 \times k$  vector, type

```
estat expenditure [if][in], atmeans
```

With this syntax, estat expenditure computes the means of all the price, expenditure, and demographic variables of your model and computes the expenditure elasticities for the k goods at those means.

To compute the uncompensated price elasticities for individual observations in the dataset, type

```
estat uncompensated [type] \{stub* | newvar_1, \ldots, newvar_{k^2}\} [if] [in]
```

Specify a stub or specify  $k^2$  new variables. Variables are created in the order  $v_{1,1},\ldots,v_{1,k},v_{2,1},\ldots,v_{2,k},\ldots,v_{k,1},\ldots,v_{k,k}$ , where  $v_{i,j}$  represents the elasticity of good i with respect to changes in the price of good j. To avoid confusion, we strongly recommend specifying a new variable stub rather than specifying  $k^2$  new variables. estat uncompensated adds variable labels to the new variables to aid in interpretation.

To compute the uncompensated price elasticities at the means of all the variables and have the results returned in a  $k \times k$  matrix, type

```
estat uncompensated [if][in], atmeans
```

With this syntax, estat uncompensated computes a  $k \times k$  matrix. The entry in row i, column j contains the elasticity of good i with respect to the price of good j.

Compensated price elasticities are obtained similarly to uncompensated price elasticities. To obtain observation-specific compensated price elasticities, type

```
\texttt{estat} \ \underline{\texttt{compensated}} \ \left[ \ type \ \right] \ \left\{ stub* \ | \ newvar_1 \ , \ \dots, \ newvar_{k^2} \right\} \ \left[ \ if \ \right] \ \left[ \ in \ \right]
```

To compute the compensated price elasticities at the means of all the variables, type

```
estat compensated [if][in], atmeans
```

### 3.4 Saved results

quaids saves the following in e():

```
Scalars
   e(N)
                           number of observations
    e(11)
                           log likelihood
   e(N_clust)
                           number of clusters
   e(ndemos)
                           number of demographics
   e(anot)
                           value of \alpha_0
    e(ngoods)
                           number of goods
Macros
    e(cmd)
                           quaids
    e(clustvar)
                           name of cluster variable
                           vcetype specified in vce()
    e(vce)
   e(vcetype)
                           title used in label Std. Err.
   e(properties)
    e(estat_cmd)
                           program used to implement estat
   e(predict)
                           program used to implement predict
    e(demographics)
                           demographic variables included
    e(lhs)
                           expenditure share variables
    e(expenditure)
                           expenditure variable
    e(lnexpenditure)
                           log-expenditure variable
   e(prices)
                           price variables
                           log-price variables
    e(lnprices)
    e(quadratic)
                           noquadratic
Matrices
   e(b)
                           coefficient vector
                           variance-covariance matrix of the estimators
   e(V)
                           estimated \alpha vector
   e(alpha)
   e(beta)
                           estimated \boldsymbol{\beta} vector
                           estimated \gamma matrix
   e(gamma)
                           estimated \dot{\lambda} vector (if applicable)
    e(lambda)
   e(eta)
                           estimated \eta matrix (if applicable)
                           estimated \rho vector (if applicable)
    e(rho)
Functions
                           marks estimation sample
    e(sample)
```

estat expenditure, atmeans saves r(expelas), the vector of expenditure elasticities.

estat uncompensated, atmeans saves r(uncompelas), the matrix of uncompensated price elasticities. The i, j entry is the elasticity of good i with regard to price j.

estat compensated, atmeans saves r(compelas), the matrix of compensated price elasticities. The i, j entry is the elasticity of good i with regard to price j.

# 4 Example

Here we illustrate the quaids command using the same dataset used in Poi (2002b), Poi (2008), and [R] **nlsur**. We first create a random integer representing the number of children in each household and a random binary variable representing rural versus urban households so that we can demonstrate a model that includes demographics. We then fit a quadratic AIDS model using  $\alpha_0 = 10$ . To save space, we use the nolog option to suppress the iteration log.

```
. webuse food
. set seed 1
. generate nkids = int(runiform()*4)
. generate rural = (runiform() > 0.7)
. quaids w1-w4, anot(10) prices(p1-p4) expenditure(expfd)
> demographics(nkids rural) nolog
(obs = 4048)
Calculating NLS estimates...
Calculating FGNLS estimates...
FGNLS iteration 2...
FGNLS iteration 3...
FGNLS iteration 4...
```

#### Quadratic AIDS model

Number of obs = Number of demographics = Alpha\_0 = Log-likelihood = 13098.966

|             | Coef.    | Std. Err. | z     | P> z  | [95% Conf. | Interval] |
|-------------|----------|-----------|-------|-------|------------|-----------|
| alpha       |          |           |       |       |            |           |
| alpha_1     | .910531  | .0696028  | 13.08 | 0.000 | .7741121   | 1.04695   |
| alpha_2     | 1328957  | .0707108  | -1.88 | 0.060 | 2714863    | .0056949  |
| alpha_3     | .0155448 | .0419878  | 0.37  | 0.711 | 0667499    | .0978394  |
| alpha_4     | .2068199 | .070847   | 2.92  | 0.004 | .0679623   | .3456775  |
| beta        |          |           |       |       |            |           |
| beta_1      | .2130298 | .0294805  | 7.23  | 0.000 | .1552492   | .2708105  |
| beta_2      | 1323188  | .0277094  | -4.78 | 0.000 | 1866283    | 0780093   |
| beta_3      | 0347438  | .015612   | -2.23 | 0.026 | 0653428    | 0041449   |
| beta_4      | 0459672  | .027198   | -1.69 | 0.091 | 0992743    | .0073399  |
| gamma       |          |           |       |       |            |           |
| $gamma_1_1$ | .2800049 | .046504   | 6.02  | 0.000 | .1888587   | .3711511  |
| gamma_2_1   | 1508244  | .0305337  | -4.94 | 0.000 | 2106693    | 0909794   |
| gamma_3_1   | 0607434  | .0133462  | -4.55 | 0.000 | 0869016    | 0345853   |
| $gamma_4_1$ | 0684371  | .0235138  | -2.91 | 0.004 | 1145233    | 022351    |
| gamma_2_2   | .1262665 | .0263559  | 4.79  | 0.000 | .0746098   | .1779231  |
| gamma_3_2   | .0145941 | .0075466  | 1.93  | 0.053 | 000197     | .0293852  |
| gamma_4_2   | .0099638 | .0118494  | 0.84  | 0.400 | 0132605    | .0331881  |
| gamma_3_3   | .0465848 | .0041107  | 11.33 | 0.000 | .0385279   | .0546416  |
| gamma_4_3   | 0004354  | .0044489  | -0.10 | 0.922 | 0091552    | .0082844  |
| gamma_4_4   | .0589087 | .0096251  | 6.12  | 0.000 | .0400439   | .0777736  |
| lambda      |          |           |       |       |            |           |
| lambda_1    | .0179386 | .0029482  | 6.08  | 0.000 | .0121603   | .0237169  |
| lambda_2    | 0097531  | .0026165  | -3.73 | 0.000 | 0148814    | 0046248   |
| lambda_3    | 0032877  | .0014164  | -2.32 | 0.020 | 0060638    | 0005117   |
| lambda_4    | 0048978  | .0025236  | -1.94 | 0.052 | 0098439    | .0000483  |
| eta         |          |           |       |       |            |           |
| eta_nkids_1 | .0002252 | .000316   | 0.71  | 0.476 | 0003941    | .0008446  |
| eta_nkids_2 | 0004756  | .0003542  | -1.34 | 0.179 | 0011699    | .0002186  |
| eta_nkids_3 | .0000165 | .0001179  | 0.14  | 0.889 | 0002147    | .0002476  |
| eta_nkids_4 | .0002339 | .0002466  | 0.95  | 0.343 | 0002494    | .0007173  |
| eta_rural_1 | 0010496  | .0007044  | -1.49 | 0.136 | 0024301    | .000331   |
| eta_rural_2 | .0006118 | .0008155  | 0.75  | 0.453 | 0009866    | .0022102  |
| eta_rural_3 | .0000242 | .0002797  | 0.09  | 0.931 | 0005239    | .0005724  |
| eta_rural_4 | .0004135 | .0005931  | 0.70  | 0.486 | 0007489    | .001576   |
| rho         |          |           |       |       |            |           |
| rho_nkids   | 1318313  | .0465259  | -2.83 | 0.005 | 2230204    | 0406422   |
| rho_rural   | .2314054 | .1909003  | 1.21  | 0.225 | 1427523    | .6055631  |

The header of the output indicates the type of model fit, the number of observations and demographic variables, the value of  $\alpha_0$  used, and the maximized value of the log-likelihood function assuming multivariate normal errors. The table of estimated parameters is divided into groups representing each vector or matrix that appears in the

demand system. Inspection reveals that the estimated parameters satisfy the adding-up and homogeneity conditions.

Commands like test and testn1 can be used to perform Wald tests on the parameters just like any other estimation command. Here we test the null hypothesis that the dummy variable rural plays no significant role in determining expenditure patterns. If that null hypothesis is true, then all elements of the row of the  $\eta$  matrix corresponding to rural must be jointly 0, along with the corresponding element of the  $\rho$  vector.

```
. test [eta]_b[eta_rural_1], notest
( 1) [eta]eta_rural_1 = 0
. test [eta]_b[eta_rural_2], notest accumulate
(1)
      [eta]eta_rural_1 = 0
(2)
      [eta]eta\_rural\_2 = 0
. test [eta]_b[eta_rural_3], notest accumulate
(1) [eta]eta rural 1 = 0
     [eta]eta\_rural\_2 = 0
( 3) [eta]eta_rural_3 = 0
. test [eta]_b[eta_rural_4], notest accumulate
( 1) [eta]eta_rural_1 = 0
      [eta]eta_rural_2 = 0
      [eta]eta_rural_3 = 0
(3)
(4)
      [eta]eta\_rural\_4 = 0
. test [rho]_b[rho_rural], accumulate
( 1) [eta]eta_rural_1 = 0
(2)
      [eta]eta_rural_2 = 0
( 3) [eta]eta_rural_3 = 0
     [eta]eta\_rural\_4 = 0
(4)
(5)
     [rho]rho_rural = 0
      Constraint 4 dropped
          chi2(4) =
                         3.42
        Prob > chi2 =
                         0.4909
```

The output from test indicates that Constraint 4 dropped. Recall that the rows of  $\eta$  are constrained to sum to 0. If the first three elements of the row corresponding to rural are each equal to 0, then the fourth must be 0 as well. Therefore, testing whether all four elements are equal to 0 is equivalent to testing whether the first three are equal to 0. In fact, knowing that, we could have omitted the line

```
. test [eta]_b[eta_rural_4], notest accumulate
```

We would have obtained the same test statistic either way. Given that our rural variable was randomly generated, we are not surprised that the test statistic does not allow us to reject the null hypothesis.

We are often more interested in the expenditure and price elasticities rather than in the estimated coefficients per se. Here we compute the expenditure elasticities for each household in the dataset and then summarize them:

- . estat expenditure e\*
- . summarize e\_1-e\_4

| Variable | Obs  | Mean     | Std. Dev. | Min       | Max      |
|----------|------|----------|-----------|-----------|----------|
| e_1      | 4039 | 1.966449 | 1.399172  | 1.261789  | 59.43863 |
| e_2      | 4029 | 0024687  | 1.061976  | -31.56926 | .7976551 |
| e_3      | 3996 | .2458694 | .9005266  | -22.42106 | .9050692 |
| e_4      | 4047 | .631619  | .8235479  | -45.74813 | .9296512 |

Recall that (3) includes  $w_i$  in the denominator of a fraction. Thus if a household does not consume a particular item ( $w_i = 0$ ), the expenditure elasticity will be infinite (a missing value in Stata parlance). That explains why the number of nonmissing observations for each of our four expenditure elasticities is less than the estimation sample size. Moreover, if  $w_i$  is close to 0, then the expenditure elasticity will be very large in magnitude. When computing household-level elasticities, you should therefore be cognizant of extreme values for some households and perhaps use summary statistics such as medians that are not influenced by outliers.

Next we compute the uncompensated price elasticities for "representative" rural and urban households, where we set all variables other than rural equal to their group-level means. To do that, we use the atmeans option of estat uncompensated, and we use an if condition to restrict computations to the relevant subsample.

```
. estat uncompensated if rural, atmeans
. matrix uprural = r(uncompelas)
. estat uncompensated if !rural, atmeans
. matrix upurban = r(uncompelas)
. matrix list uprural
uprural[4,4]
            c1
   -1.2651808 -.10694767
                            -.13132521
                                        -.25314115
r1
    .32700506 -.74128023
                                         .12540763
r2
                            .04191575
                            -.55763258
      .0511128 -.03848514
                                         .05041919
     .08810754 -.06235336 -.00975406
                                        -.73724268
. matrix list upurban
upurban[4,4]
            c1
                                                c4
r1
    -1.3038599
                -.0934749
                             -.1296122
                                        -.25505899
     .36542124 -.75058651
r2
                             .04011097
                                         .12743289
                            -.55976996
r3
     .07559882
                -.04814826
                                         .05029575
      .1021857 -.06701876
                            -.01060621
r4
                                        -.74324303
```

The entry in row i, column j of each elasticity matrix represents the percentage change in the quantity of good i consumed for a 1% change in the price of good j. Among rural consumers, a 1% increase in the price of good A raises consumption of good B by 0.33%.

# 5 Conclusion

Previously, to fit an AIDS model in Stata, a user would have to write his or her own program to be used with nlsur; this proved a challenge for beginners. Widely distributed example programs did not allow for demographic variables either. The new quaids command removes all programming-related barriers to fitting these models in Stata, and it allows one to include demographic variables.

Even if users were successful in fitting an AIDS model, many probably struggled with computing elasticities; postestimation commands after quaids make those tasks easy.

## 6 References

- Banks, J., R. Blundell, and A. Lewbel. 1997. Quadratic Engel curves and consumer demand. Review of Economics and Statistics 79: 527–539.
- Blacklow, P., A. Nicholas, and R. Ray. 2010. Demographic demand systems with application to equivalence scales estimation and inequality analysis: The Australian evidence. *Australian Economic Papers* 49: 161–179.
- Deaton, A. S., and J. Muellbauer. 1980a. *Economics and Consumer Behavior*. Cambridge: Cambridge University Press.
- ———. 1980b. An almost ideal demand system. American Economic Review 70: 312–326.
- Mas-Colell, A., M. D. Whinston, and J. R. Green. 1995. *Microeconomic Theory*. New York: Oxford University Press.
- Poi, B. P. 2002a. Dairy policy and consumer welfare. In Three Essays in Applied Econometrics, Chapter II, Doctoral thesis. Department of Economics, University of Michigan.
- ———. 2002b. From the help desk: Demand system estimation. Stata Journal 2: 403–410.
- ———. 2008. Demand-system estimation: Update. Stata Journal 8: 554–556.
- Pollak, R. A., and T. J. Wales. 1981. Demographic variables in demand analysis. *Econometrica* 49: 1533–1551.
- Ray, R. 1983. Measuring the costs of children: An alternative approach. Journal of Public Economics 22: 89–102.

#### About the author

Brian Poi is a senior economist at StataCorp LP. He is indebted to Kompal Sinha and Helder Zavale for valuable feedback on an earlier draft of this article and the quaids command.